

mC gramatika

Simboli

letter

→ "a" | "A" | "b" | "B" | "c" | "C" | "d" | "D" | "e" | "E" | "f" | "F" | "g" | "G" | "h" | "H"
| "i" | "I" | "j" | "J" | "k" | "K" | "l" | "L" | "m" | "M" | "n" | "N" | "o" | "O" | "p" | "P"
| "q" | "Q" | "r" | "R" | "s" | "S" | "t" | "T" | "u" | "U" | "v" | "V" | "w" | "W" | "x" | "X"
| "y" | "Y" | "z" | "Z"

digit

→ "0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9"

identifier

→ letter (letter | digit)*

int_constant

→ digit +

Pojmovi

program

→ *function_list*

function_list

→ *function*

→ *function_list function*

function

→ *type* identifier "(" ")" *body*

type

→ "int"

body

→ "{" *variable_list* *statement_list* "}"

→ "{" *statement_list* "}"

variable_list

→ *variable* ";"

→ *variable_list* *variable* ";"

variable

→ *type* identifier

statement_list

→ ε

→ *statement_list* *statement*

```

statement
  → compound_statement
  → assignment_statement
  → if_statement
  → return_statement

compound_statement
  → "{" statement_list "}"

assignment_statement
  → identifier "=" num_exp ";"

num_exp
  → exp
  → num_exp "+" exp
  → num_exp "-" exp

exp
  → constant
  → identifier
  → function_call
  → "(" num_exp ")"

constant
  → int_constant

function_call
  → identifier "(" ")"

if_statement
  → if_part
  → if_part "else" statement

if_part
  → "if" "(" rel_exp ")" statement

rel_exp
  → num_exp "==" num_exp
  → num_exp "!=" num_exp

return_statement
  → "return" num_exp ";"

```

Komentari:

višelinijski: /* a comment */

jednolinijski: // a comment

Razmak i kraj linije imaju funkciju separatora simbola